

## Handout 5

### Answer Set Programming (II)

A *rule* is an implication of the form

$$F \leftarrow G \tag{1}$$

where  $F$  and  $G$  are formulas that do not contain  $\rightarrow$  and  $\leftrightarrow$ .  $F$  is called the *head* of the rule and  $G$  is called the *body*. If  $G$  is  $\top$  we often identify the rule with its head by dropping  $\leftarrow \top$ . If  $F$  is  $\perp$  we often write the rule as  $\leftarrow G$ .

A *program* is a finite set of rules.

The answer sets of a program are defined as follows. The *reduct*  $\Pi^X$  of a program  $\Pi$  relative to a set  $X$  of atoms is obtained from  $\Pi$  by replacing all maximal occurrences of  $\neg H$  where  $H$  is a formula by

- $\perp$  if  $X \models H$ ,
- $\top$  otherwise.

We say that  $X$  is an *answer set* of  $\Pi$  if  $X$  is a minimal set satisfying  $\Pi^X$ . The minimality of  $X$  is understood here in the sense of set inclusion.

**5.1<sup>c</sup>** (a) Program  $\neg\neg p$  has no answer sets. (b) Each of the two programs

$$p \vee \neg p \tag{2}$$

and

$$p \leftarrow \neg\neg p \tag{3}$$

has two answer sets:  $\emptyset$  and  $\{p\}$ .

### Nonmonotonicity of Answer Set Semantics

It is easy to see that any answer set of program  $\Pi$  satisfies  $\Pi$ . (prove it!) In this sense, the answer set semantics is stronger than the classical semantics, which is given by the concept of satisfaction. The sets satisfying a formula are called its “models,” and the answer sets of a formula are called its “stable” models.

The classical semantics of propositional formulas is monotonic, in the sense that conjoining a formula with another formula can only make the set of its models smaller: if  $X$  satisfies  $F \wedge G$  then  $X$  satisfies  $F$ . The answer set semantics is nonmonotonic: an answer set of  $F \wedge G$  does not have to be an answer set of  $F$ . This phenomenon is observed even when  $F$  and  $G$  are positive programs, and it is related to the minimality condition in the definition of an answer set. For instance, the answer set  $\{p, q\}$  of  $p \wedge q$  is not an answer set of  $p$ . The usefulness of nonmonotonicity as a property of knowledge representation formalisms and its relation to minimality conditions were established in early work on circumscription [McCarthy, 1980].

## 5.2<sup>c</sup>

(a) Program

$$p \vee q$$

has two answer sets:  $\{p\}$  and  $\{q\}$ .

(b) Program

$$p \vee q$$

$$p \leftarrow q$$

$$q \leftarrow p$$

has one answer set:  $\{p, q\}$ .

These facts provide one more example of nonmonotonicity: after appending the last two rules in (b), the program  $p \vee q$  gets a new answer set. They show also that disjunction, under the answer set semantics, is sometimes “exclusive,” and sometimes not. On the one hand, among the three sets

$$\{p\}, \{q\}, \{p, q\}$$

that satisfy  $p \vee q$ , the third is not an answer set. On the other hand, appending the last two rules to the disjunction makes  $\{p, q\}$  an answer set; the disjunction “becomes inclusive.”

## Choice Formulas and Constraints

The art of answer set programming is based on the possibility of representing the collection of sets that we are interested in as the collection of answer sets of a logic program. This is often achieved by combining rules of two kinds, studied in this handout. A “choice rule” is a program with many answer

sets that are more than the collection of sets that we want to describe. The additional answer sets can be removed by adding “constraints” to this program.

### Choice Formulas

For any finite set  $Z$  of atoms, by  $Z^c$  we denote the program

$$\bigwedge_{A \in Z} (A \vee \neg A)$$

**5.3<sup>c</sup>** Prove the following statement: A set  $X$  is an answer set of  $Z^c$  iff  $X$  is a subset of  $Z$ .

Thus if  $Z$  consists of  $n$  atoms then  $Z^c$  has  $2^n$  answer sets. Under the answer set semantics,  $Z^c$  says: choose for every element of  $Z$  arbitrarily whether to include it in the answer set. We will call formulas of the form  $Z^c$  *choice formulas*. (The superscript  $c$  is used in this notation because it is the first letter of the word “choice.”)

Although LPARSE does not allow us to use conjunctions and disjunctions in the heads of rules, it does understand choice formulas as heads, with the superscript  $c$  dropped. For instance, we can write

$$\{p, q\}$$

for

$$(p \vee \neg p) \wedge (q \vee \neg q)$$

and

$$\{p\} :- q$$

for

$$p \vee \neg p \leftarrow q.$$

**5.4<sup>c</sup>** (a) Consider the program

$$\begin{aligned} &\{p, q, r\}^c \\ &s \leftarrow p, \neg q. \end{aligned}$$

What do you think its answer sets are? Use SMOELS to verify your conjecture. (b) Do the same for the program

$$\begin{aligned} &\{p, q\}^c \\ &\{r, s\}^c \leftarrow \neg p. \end{aligned}$$

If the head of a rule is a choice formula  $Z^c$  for a large set  $Z$  then it may be possible to represent the rule in the language of LPARSE concisely using variables (see Handout 4 on the use of variables in LPARSE). For instance, the rule

$$\{p_1, \dots, p_7\}^c \leftarrow q \quad (4)$$

can be encoded as follows:

```
index(1..7).
{p(I) : index(I)} :- q.
```

Note how the “local” use of  $I$  in this example differs from the “global” use of variables in Handout 4 and in the following example:

```
index(1..7).
{p(I)} :- q, index(I).
```

The last two lines correspond to the set of 7 rules:

$$\{p_i\}^c \leftarrow q \quad (1 \leq i \leq 7). \quad (5)$$

In this example, the difference between local and global variables is not essential, however: there is a theory that tells that replacing (4) with (5) in any program does not affect the program’s answer sets.<sup>1</sup>

## Constraints

A *constraint* is a rule with the head  $\perp$ , that is to say, a rule of the form  $\perp \leftarrow F$ . (Recall it can be abbreviated as  $\leftarrow F$ .) The following theorem tells us how adding a constraint to a program affects the collection of its answer sets.

**Theorem on Constraints** [Lifschitz *et al.*, 1999]. For any program  $\Pi$  and formula  $F$ , a set  $X$  of atoms is an answer set for  $\Pi \cup \{\leftarrow F\}$  iff  $X$  is an answer set for  $\Pi$  and does not satisfy  $F$ .

**5.5<sup>c</sup>** Find the answer sets of the program

$$\begin{array}{l} \{p, q, r\}^c \\ \leftarrow p, q, \neg r \end{array}$$

(a) using Theorem on Constraints; (b) using SMOLELS.

---

<sup>1</sup>This theory is called “strong equivalence,” which is beyond the scope of this class.

We observed earlier that a program  $\Pi_1 \cup \Pi_2$  may have answer sets that are not found among the answer sets of  $\Pi_1$ . The assertion of Problem 5 shows, however, that this cannot happen if  $\Pi_2$  is a constraint. Conjoining a program with a constraint leads only to the loss of answer sets.

The combination of choice rules and constraints yields a way to embed propositional logic into the answer set semantics as follows.

**5.6** For any propositional formula  $F$ , a set  $X$  of atoms is a model of  $F$  iff  $X$  is an answer set of

$$\begin{array}{l} Z^c \\ \leftarrow \neg F \end{array}$$

where  $Z$  is the set of all atoms occurring in  $F$ .

## Cardinality Expressions

Constraints used in ASP programs often involve conditions on the cardinality of a set of atoms. We will introduce special notation for such formulas.

For any finite set  $Z$  of literals and any nonnegative integer  $l$  (“lower bound”), by  $l \leq Z$  we denote the disjunction of the formulas  $\bigwedge_{L \in Y} L$  over all  $l$ -element subsets  $Y$  of  $Z$ . For instance,

$$2 \leq \{p, q, r\}$$

stands for

$$(p \wedge q) \vee (p \wedge r) \vee (q \wedge r).$$

Clearly, if the elements of  $Z$  are atoms then a subset  $X$  of  $Z$  satisfies  $l \leq Z$  iff the cardinality of  $X$  is at least  $l$ .

By  $Z \leq u$ , where  $u$  is a nonnegative integer (“upper bound”), we denote the formula  $\neg(u + 1 \leq Z)$ . Finally,  $l \leq Z \leq u$  stands for

$$(l \leq Z) \wedge (Z \leq u).$$

As an example of the use of cardinality expressions in constraints, consider the program

$$\begin{array}{l} \{p_1, \dots, p_n\}^c \\ \leftarrow \{p_1, \dots, p_n\} \leq 0 \\ \leftarrow 2 \leq \{p_1, \dots, p_n\}. \end{array} \tag{6}$$

The constraints eliminate two kinds of sets from the collection of answer sets of the choice rule: the empty set and the sets that have at least 2 elements. Consequently, the answer sets of (6) are the singletons  $\{p_1\}, \dots, \{p_n\}$ .

The language of LPARSE allows us to use cardinality expressions

$$l \leq Z, Z \leq u, l \leq Z \leq u$$

in the bodies of rules, with the sign  $\leq$  dropped. For instance, program (6) can be written as

```
index(1..n).

{p(I) : index(I)}.

:- {p(I) : index(I)} 0.
:- 2 {p(I) : index(I)}.
```

**5.7<sup>c</sup>** Although the syntax of LPARSE does not allow us, generally, to use nested negations, the occurrence of formula  $\neg\neg p$  can sometimes be written as a cardinality expression that LPARSE understands. Find such an expression and use SMODELS to find the answer sets

$$p \leftarrow \neg\neg p. \quad (7)$$

If  $Z$  is a finite set of atoms, we will use

$$\begin{aligned} l \leq Z^c & \text{ as shorthand for } Z^c \wedge (l \leq Z), \\ Z^c \leq u & \text{ as shorthand for } Z^c \wedge (Z \leq u), \\ l \leq Z^c \leq u & \text{ as shorthand for } Z^c \wedge (l \leq Z \leq u). \end{aligned}$$

**Proposition on cardinality expression** *For any pairwise distinct atoms  $A_1, \dots, A_n$ , nonnegative integers  $l$  and  $u$ , and a set  $X$  of atoms,*

- (i)  *$X$  is an answer set of  $l \leq \{A_1, \dots, A_n\}^c$  iff  $X \subseteq \{A_1, \dots, A_n\}$  and  $l \leq |X|$ ;*
- (ii)  *$X$  is an answer set of  $\{A_1, \dots, A_n\}^c \leq u$  iff  $X \subseteq \{A_1, \dots, A_n\}$  and  $|X| \leq u$ ;*
- (iii)  *$X$  is an answer set of  $l \leq \{A_1, \dots, A_n\}^c \leq u$  iff  $X \subseteq \{A_1, \dots, A_n\}$  and  $l \leq |X| \leq u$ .*

The language of LPARSE allows us to use expressions

$$l \leq Z^c, Z^c \leq u, l \leq Z^c \leq u$$

in heads of rules, with both  $\leq$  and  $^c$  dropped. For instance, program

$$1 \leq \{p_1, \dots, p_n\}^c \leq 1. \quad (8)$$

can be written as

`index(1..n).`

`1 {p(I) : index(I)} 1.`

Note that LPARSE understands the expression

`1 {...} u`

in different ways depending on whether it occurs in the body or in the head of a rule: it stands for

$$l \leq \{\dots\} \leq u$$

in the body, and for

$$l \leq \{\dots\}^c \leq u$$

in the head.

**5.8<sup>c</sup>** Consider the program

$$1 \leq \{p_{i1}, \dots, p_{in}\}^c \leq 1 \quad (1 \leq i \leq n),$$

where  $n$  is a positive integer. How many answer sets does this program have, in your opinion? Check your conjecture for  $n = 3$  using SMOELS.

## Methodology of Answer Set Programming

Recall that to solve a problem using ASP means to write a logic program whose answer sets correspond to solutions, and then find solutions using an answer set solver. The basic approach to writing such programs consists in combining choice rules with constraints. In this section we discuss a few examples of this “generate-and-test” strategy and its enhancement that involves “defined” atoms.

### Example: N Queens

Our goal is to place  $n$  queens on an  $n \times n$  chessboard so that no two queens would be placed on the same row, column or diagonal. A solution can be described by a set of atoms of the form  $q(i, j)$  ( $1 \leq i, j \leq n$ ); including  $q(i, j)$  in the set indicates that there is a queen at position  $(i, j)$ . A solution is a set  $X$  satisfying the following conditions:

1. The cardinality of  $X$  is  $n$ .

2.  $X$  does not contain a pair of different atoms of the form  $q(i, j), q(i', j)$  (two queens on the same column).
3.  $X$  does not contain a pair of different atoms of the form  $q(i, j), q(i, j')$  (two queens on the same row).
4.  $X$  does not contain a pair of different atoms  $q(i, j), q(i', j')$  with  $|i' - i| = |j' - j|$  (two queens on the same diagonal).

The sets satisfying Conditions 1 and 2 can be described by rules similar to the program from Problem 4.5:

$$1 \leq \{q(1, j), \dots, q(n, j)\}^c \leq 1 \quad (1 \leq j \leq n)$$

(exactly one queen on each column). These rules form the “generate” part of our program. The “test” part consists of the constraints expressing Condition 3

$$\leftarrow q(i, j), q(i, j') \quad (1 \leq i, j, j' \leq n; j < j')$$

and Condition 4

$$\leftarrow q(i, j), q(i', j') \quad (1 \leq i, i', j, j' \leq n; j < j'; |i' - i| = j' - j).$$

Here is a representation of this program in the input language of LPARSE:

```
number(1..n).
#domain number(I;I1;J;J1).

1{q(K,J) : number(K)}1.

:- q(I,J), q(I,J1), J<J1.

:- q(I,J), q(I1,J1), J<J1, abs(I1-I)==J1-J.
```

The expression `number(I;I1;J;J1)` is the LPARSE abbreviation for the list

```
number(I), number(I1), number(J), number(J1).
```

**5.9<sup>c</sup>** Use SMODELS to find all solutions to the 8 queens problem that (a) have a queen at (1,1); (b) have no queens in the  $4 \times 4$  square in the middle of the board.



**5.10<sup>c</sup>** (a) Check how long it takes for SMOLELS to find one solution to the 16 queens problem using the program above. (b) The constraint corresponding to Condition 3 can be alternatively written using a cardinality expression:

$$\leftarrow 2 \leq \{q(i, 1), \dots, q(i, n)\} \quad (1 \leq i \leq n).$$

Modify the program accordingly and check how this modification affects the computation time for 16 queens.

### Example: Schur Numbers

We will show now how to use ASP to estimate the size of Schur numbers. A set  $A$  of integers is called *sum-free* if there are no numbers  $x, y$  in  $A$  such that  $x + y$  is in  $A$  also ( $x$  and  $y$  do not need to be different). The *Schur number*  $S(k)$  is the largest integer  $n$  for which the interval  $\{1, \dots, n\}$  can be partitioned into  $k$  sum-free sets.<sup>2</sup>

For specific values of  $k$ ,  $S(k)$  can be computed (or estimated) using a program whose answer sets correspond to the partitions of  $\{1, \dots, n\}$  into  $k$  (possibly empty) sum-free sets. We will use the atoms  $a_i(x)$  ( $1 \leq i \leq k$ ,  $1 \leq x \leq n$ ) to express that  $x$  belongs to the  $i$ -th set  $A_i$ . The choice rules

$$1 \leq \{a_1(x), \dots, a_k(x)\} \leq 1 \quad (1 \leq x \leq n)$$

describe “potential solutions”—partitions of  $\{1, \dots, n\}$  into  $k$  sets. The constraints

$$\leftarrow a_i(x), a_i(y), a_i(x + y) \quad (x, y \geq 1, x + y \leq n, 1 \leq i \leq k)$$

eliminate the sets  $A_i$  that are not sum-free.

Here is how this program can be written in the input language of LPARSE:

```
number(1..n).
#domain number(X;Y).

subset(1..k).

1{a(I,X) : subset(I)}1.

:- a(I,X), a(I,Y), a(I,X+Y), subset(I), X+Y<=n.
```

---

<sup>2</sup>See <http://mathworld.wolfram.com/SchurNumber.html> .

**5.11<sup>c</sup>** Use this program to find  $S(3)$ .

The number of answer sets for the ASP program above is greater than the number of partitions of  $\{1, \dots, n\}$  into  $k$  sum-free sets, because the sets in any such partition can be arranged in a sequence  $A_1, \dots, A_k$  in many possible ways. In particular, the subsets in a partition can be always ordered in such a way that

$$\begin{aligned} 1 &\in A_1, \\ 2 &\in A_1 \cup A_2, \\ &\dots \\ k &\in A_1 \cup \dots \cup A_k. \end{aligned}$$

Consequently, if we add to the program above a constraint expressing these conditions then the property of the program that we are interested in will be preserved—it will have an answer set iff  $\{1, \dots, n\}$  can be partitioned into  $k$  sum-free sets. Adding this “symmetry breaking” constraint will actually improve the computation time of SMOODELS for many values of  $k$  and  $n$ , because, generally, SMOODELS is good at exploiting constraints for restricting search.

**5.12<sup>c</sup>** Investigate the effect of this optimization on the computation time of SMOODELS for  $k = 4$ ,  $n = 38$ .

## Generate-Define-Test

In more complex uses of ASP, constraints in the test part of the program use auxiliary atoms that are defined in terms of the atoms occurring in the generate part. The definitions of the auxiliary atoms form then a third component of the program, besides the “generate” and “test” parts—its “define” part.

Consider, for instance, the problem of finding a Hamiltonian circuit in a given graph, that is, a closed path that visits every vertex of the graph exactly once. A Hamiltonian circuit in a graph  $G$  can be thought of as a subgraph  $C$  of  $G$  that has the same set  $V$  of vertices as  $G$  and satisfies two conditions:

1. Every vertex is adjacent in  $C$  to exactly two vertices.
2. Every vertex is reachable in  $C$  from some fixed vertex  $u_0$ .

The program below represents an edge  $\{u, v\}$  of  $G$  by the atom  $in(u, v)$ , where  $u < v$  ( $<$  is a fixed total order on  $V$ ). The presence of this atom in

an answer set indicates that the edge  $\{u, v\}$  is included in  $C$ . The generate part of the program consists of the rules

$$\{in(u, v)\}^c \quad (9)$$

for all edges  $\{u, v\}$  of  $G$  ( $u < v$ ). Thus any set of edges is a potential solution. To encode the two conditions that characterize Hamiltonian circuits, we introduce the auxiliary atoms  $adj(u, v)$  ( $u < v$ ) that represent adjacency in  $C$ . These atoms are “defined” in terms of  $in(u, v)$  by the rules

$$adj(u, v), adj(v, u) \leftarrow in(u, v) \quad (10)$$

for all edges  $\{u, v\}$  of  $G$  ( $u < v$ ). Then Condition 1 above can be expressed by the constraints

$$\begin{aligned} &\leftarrow \{adj(u, v) : v \in V\} \leq 1 \\ &\leftarrow 3 \leq \{adj(u, v) : v \in V\} \end{aligned} \quad (11)$$

for all  $u \in V$ . For Condition 2, we introduce the auxiliary atoms  $r(u)$  that express the reachability of  $u$  from  $u_0$ . They are “defined recursively” by the rules

$$\begin{aligned} &r(u_0) \\ &r(u) \leftarrow r(v), adj(u, v) \end{aligned} \quad (12)$$

for all edges  $\{u, v\}$  of  $G$ . Using these atoms, we express Condition 2 by the constraints

$$\leftarrow not\ r(u) \quad (13)$$

for all  $u \in V$ .

Answer sets for program (9)–(13) are in a 1–1 correspondence with the Hamiltonian circuits in  $G$ . Rules (10) and (12) form the define part of the program, and rules (11) and (13) form the test part.

The following file `hc` is an LPARSE encoding of program (9)–(13):

```
{in(U,V)} :- edge(U,V).

adj(U,V) :- in(U,V), vertex(U;V).
adj(V,U) :- in(U,V), vertex(U;V).

:- {adj(U,V) : vertex(V)}1, vertex(U).
:- 3{adj(U,V) : vertex(V)}, vertex(U).

r(0).
r(U) :- r(V), adj(U,V), vertex(U;V).
```

```

:- not r(U), vertex(U).

hide.
show in(_,_).

```

The last two lines tell SMOBELS to display the `in` atoms only. They make the use of the option `-d none` to suppress information on the domain predicates redundant

This file needs to be appended to a description of  $G$  in the form of a definition of the domain predicates `vertex` and `edge`. For instance, the file

```

vertex(u0;u1;u2;u3;v0;v1;v2;v3).
edge(u0,u1). edge(u1,u2). edge(u2,u3). edge(u3,u0).
edge(v0,v1). edge(v1,v2). edge(v2,v3). edge(v3,v0).
edge(u0,v0). edge(u1,v1). edge(u2,v2). edge(u3,v3).

```

describes the vertices and edges of a cube. Assuming that this file is called `cube`, SMOBELS can be instructed to find a Hamiltonian circuit by the command

```
% lparse cube hc | smodels
```

**5.13<sup>c</sup>** The vertices of graph  $G_n$  are the points  $(x, y)$  with coordinates  $x, y \in \{0, \dots, n-1\}$ ; two vertices are adjacent if the distance between them is 1. Use SMOBELS to find out whether  $G_5$  has a Hamiltonian circuit.

Syntactically, rules in the generate and test parts of an ASP program are somewhat different from Prolog rules: Prolog has neither choice rules nor constraints. On the other hand, the define part of an ASP program does look similar to a Prolog program, and writing define rules in ASP is based on the same intuition as writing Prolog programs.

For “generate-and-test” programs, such as the examples discussed in Sections and , proving correctness is usually not difficult: it requires no theory of answer sets beyond the basic properties of choice formulas and constraints discussed in Handout 4. circuit example, then more complex mathematical tools are required for a correctness proof.

## Programming Exercises

In each of the following exercises, show how to solve the given computational problem using SMOBELS. Test your programs using data of your choice if data is not given.

**5.14<sup>c</sup>** Sudoku (a.k.a. Number Place) is a game to fill in the grid so that every row, every column, and every  $3 \times 3$  box contains the digits 1 through 9. The following is an instance of Sudoku.

	6		1		4		5	
		8	3		5	6		
2								1
8			4		7			6
		6				3		
7			9		1			4
5								2
		7	2		6	9		
	4		5		8		7	

Problem: fill in the numbers.

**5.15<sup>c</sup>** Each of four men owns a different species of exotic pet. Here is what we know about them:

1. Mr Engels (whose pet is named Sparky), Abner and Mr. Foster all belong to a club for owners of unusual pets.
2. The iguana isn't owned by either Chuck or Duane.
3. Neither the jackal nor the king cobra is owned by Mr. Foster.
4. The llama doesn't belong to Duane (whose pet is named Waggles).
5. Abner, who doesn't own the king cobra, isn't Mr. Gunter.
6. Bruce and Mr. Foster are neighbors.
7. Mr. Halevy is afraid of iguanas.

Problem: Find each man's full name and determine what kind of pet he owns.

**5.16<sup>c</sup>** There are five houses of five different colors. In each house lives a person of a different nationality. Each of these five men drinks a certain beverage, smokes a certain brand of cigarettes, and keeps a certain pet. No two men have the same pet, drink the same drink or smoke the same brand. We also know the following:

1. The Brit lives in the red house.

2. The Swede keeps a dog.
3. The Dane drinks tea.
4. The green house is on the left of the white house.
5. The owner of the green house drinks coffee.
6. The person who smokes Pall Mall rears birds.
7. The owner of the yellow house smokes Dunhill.
8. The man living in the house right in the center drinks milk.
9. The Norwegian lives in the first house.
10. The man who smokes Blend lives next to the one who has cats.
11. The man who has horses lives next to the Dunhill smoker.
12. The man who smokes Bluemaster drinks beer.
13. The German smokes Princess.
14. The Norwegian lives next to the blue house.
15. The man who smokes Blend has a neighbor who drinks water.

Problem: determine who owns the fish.

**5.17<sup>c</sup>** An *independent set* in a graph  $G$  is a subset of its vertices in which every two elements are not adjacent. According to Ramsey's theorem, for any positive integers  $m, n$  there exists a positive integer  $r$  such that every graph with at least  $r$  vertices contains either a clique of cardinality  $m$  or an independent set of cardinality  $n$ . The smallest  $r$  with this property is called the *Ramsey number*  $R(m, n)$ .<sup>3</sup> For instance,  $R(3, 3) = 6$ . Problem: estimate  $R(m, n)$ .

**5.18<sup>c</sup>** You are organizing a large New Year's Eve party. There will be  $n$  tables in the room, with  $m$  chairs around each table. You need to select a table for each of the guests, who are assigned numbers from 1 to  $mn$ , so that two conditions are satisfied. First, some guests like each other and want to sit together; accordingly, you are given a set  $A$  of two-element subsets of  $\{1, \dots, mn\}$ , and, for every  $\{i, j\}$  in  $A$ , guests  $i$  and  $j$  should be assigned

---

<sup>3</sup>See <http://mathworld.wolfram.com/RamseyNumber.html> .

the same table. Second, some guests dislike each other and want to sit at different tables; accordingly, you are given a set  $B$  of two-element subsets of  $\{1, \dots, mn\}$ , and, for every  $\{i, j\}$  in  $B$ , guests  $i$  and  $j$  should be assigned different tables. Problem: find such a seating arrangement or determine that this is impossible.

**5.19<sup>c</sup>** You are in charge of assigning referees to the papers submitted to a conference. Each of the  $n$  submissions needs to be assigned to a few referees from among the  $m$  members of the Program Committee. The PC members have read the abstracts of all submissions, and each of them gave you the numbers of the submissions that he is qualified to referee; let  $A_i$  ( $1 \leq i \leq m$ ) be the subset of  $\{1, \dots, n\}$  given to you by the  $i$ -th committee member. You need to select, for each  $i$ , a set  $X_i$  of papers to be assigned to the  $i$ -th committee member so that three conditions are satisfied. First, each  $X_i$  should be a subset of  $A_i$ . Second, the cardinality of each  $X_i$  should be between a lower bound  $l$  and an upper bound  $u$ . Third, each paper should be assigned to exactly  $k$  referees. Problem: find such an assignment of papers to referees or determine that this is impossible.

## References

- [Lifschitz *et al.*, 1999] Vladimir Lifschitz, Lappoon R. Tang, and Hudson Turner. Nested expressions in logic programs. *Annals of Mathematics and Artificial Intelligence*, 25:369–389, 1999.
- [McCarthy, 1980] John McCarthy. Circumscription—a form of non-monotonic reasoning. *Artificial Intelligence*, 13:27–39, 171–172, 1980.